

Лабораторная работа № 5

Нелинейная динамическая контактная задача о падении массивного тела на деформируемую балку (в двумерной постановке)

КЛЮЧЕВЫЕ СЛОВА

1. Динамическая задача (нестационарный анализ)
2. Нелинейный анализ
3. Контактная задача (контактные и ответные элементы)

ОПИСАНИЕ ЗАДАЧИ

Рассмотрим нелинейную динамическую контактную задачу о падении двумерного тела в форма квадрата на деформируемую балку. Необходимо построить модель падения двумерного тела на балку. Провести расчет процесса движений тела и балки на временном промежутке, равном четырем временным промежуткам t_d , где t_d – время, за которое тело долетает до балки.

Условие задачи.

Схема задачи представлена на рис. 1. Балка длиной $l=100$ см имеет прямоугольное поперечное сечение с высотой $h_l=1.25$ см и шириной $b_l=2.56$ см. Балка выполнена из меди с модулем Юнга $E = 1.2 \cdot 10^{11}$ Н/м², коэффициентом Пуассона $\nu=0.33$, плотностью $\rho=8900$ кг/м³. Левый конец балки жестко закреплен, правый конец закреплен по оси y . На балку падает груз в форме квадрата со стороной $ab=6$ см толщины $h_a=2.5$ см, выполненный из стали с модулем Юнга $E = 2 \cdot 10^{11}$ Н/м², коэффициентом Пуассона $\nu=0.29$, плотностью $\rho=7880$ кг/м³. Первоначально груз находится на высоте $h=80$ см от балки и расположен на расстоянии $l_1=56$ см от левого конца балки.




Рис. 1

РЕШЕНИЕ ЗАДАЧИ С ИСПОЛЬЗОВАНИЕМ ANSYS

Пример решения этой задачи дан в файле Drop_20_2_h80.txt. Данный файл предназначен для выполнения в пакетном режиме ANSYS и снабжен комментариями.

Для малого шага по времени может потребоваться увеличение числа подшагов sets (по умолчанию равно 1000). Для этого в начало входного файла надо добавить строку:

/CONFIG,NRES,4000 ! 4000 будет максимальным количеством подшагов

В препроцессоре /PREP7 зададим параметры для геометрических размеров.

! Расчет падения груза на балку (2D задача)

! Все данные в системе СИ

/PREP7

/SHOW

/TITLE, Bouncing Block

! ВСЕ РАЗМЕРЫ В СИСТЕМЕ СИ

LBEAM=1 ! Длина балки

HB=0.8 ! Высота, на которой первоначально находился груз

L1=0.56 ! Расстояние от левого конца балки до груза по оси

X

BL=0.0256 ! Ширина балки

HL=0.0125 ! Толщина (высота) балки

HA=0.025 ! Толщина груза

AB=0.06 ! Сторона квадрата, занимаемого грузом

Зададим параметры для временных интервалов. Известно, что тело, находящееся на высоте h от балки упадет на нее за время $\sqrt{\frac{2h}{g}}$, где g -

ускорение свободного падения. Назовем это время расчетным временем T_DROP. Нестационарный анализ будет проведен в три этапа. На первом шаге задается очень небольшой временной интервал, чтобы рассчитать начальные условия для нестационарного анализа. Время окончания второго интервала возьмем равным расчетному времени падения груза на балку. Длительность третьего интервала может быть произвольной. Время окончания третьего интервала – это момент окончания наблюдения за грузом. Для третьего интервала потребуется задать небольшой шаг по времени, чтобы обеспечить сходимость решения.

T_DROP=SQRT(2*HB/9.8) ! Расчетное время падения груза на балку

TIME_0=0.0002 ! Время окончания первого интервала

интегрирования по времени

DTIME_0=0.0001 ! Шаг по времени для первого интервала

TIME_1=T_DROP ! Время окончания второго интервала

интегрирования по времени

TIME_2=4*T_DROP ! Время окончания третьего интервала

интегрирования по времени

DTIME_1=TIME_1/10 ! Шаг по времени для второго интервала

DTIME_2=TIME_1/50 ! Шаг по времени для третьего интервала

Будем решать задачу в двумерной постановке. Для балки выберем двумерный балочный элемент BEAM3, имеющий три степени свободы: их, uy и rotz. Балка моделируется линией, при этом ее толщина и поперечное сечение задается в наборе специальных параметров балочного конечного элемента. Для модели груза рассмотрим плоское напряженное состояние с

толщиной, возьмем элемент PLANE42 с $\text{keyopt}(3)=3$. Груз моделируется в виде квадрата, при этом толщина груза задается в наборе специальных параметров этого конечного элемента.

В стандартном контактном анализе ANSYS сначала строится конечно-элементная модель со стандартными элементами, а затем на контактных поверхностях создаются контактные и ответные элементы.

Для моделирования контакта груза с балкой определим двумерные контактные элементы CONTA175 с типом контакта узел-поверхность для поверхности груза и ответные двумерные элементы TARGE169 для балки.

```
ET,1,BEAM3 ! Балочный элемент (2D elastic beam),
степени свободы ux,uy,rotz
ET,2,PLANE42,,,3 ! Плоский четырехугольный 4-узловой КЭ
для груза с толщиной
ET,3,CONTA175 ! 2-D/3-D Node-to-Surface Contact
ET,4,TARGE169 ! Ответный (2-D TARGET) элемент (для
балки)
```

Далее зададим наборы вещественных констант. Для балочного элемента BEAM3 (тип 1) нужно задать площадь поперечного сечения балки, момент инерции и толщину балки. Для элемента PLANE42 нужно задать толщину груза. Третий набор констант будет для контактных элементов.

```
! Наборы параметров (Real constants set) для типов элементов
R,1,BL*HL, (BL*HL**3)/12,HL ! Площадь поперечного сечения, момент
инерции, толщина (для балки)
R,2,HA ! Толщина (для груза)
R,3, ! Нормальная контактная жесткость - по умолчанию
```

Далее следует определить параметры для входных данных и задать материальные свойства. Для линейной изотропного материала нужно задать модуль Юнга E , коэффициент Пуассона ν и плотность ρ . В данной задаче два разных материала, медь и сталь, поэтому два набора материальных свойств.

```
! Материальные свойства типа 1 - для балки (медь)
MP,EX,1,1.2e11 ! Модуль Юнга
MP,DENS,1,8.9e3 ! Плотность
MP,NUXY,1,0.33 ! Коэффициент Пуассона

! Материальные свойства типа 2 - для груза (сталь)
MP,EX,2,2e11 ! Модуль Юнга
MP,DENS,2,7.88e3 ! Плотность
MP,NUXY,2,0.29 ! Коэффициент Пуассона
```

В качестве параметров разбиения будем задавать число элементов на гранях груза и балки.

```
! Параметры конечно-элементной сетки
NE_BEAM=30 ! Число КЭ для балки
NE_BLOCK=2 ! Число КЭ на грани груза
```

Построим твердотельные и конечно-элементные модели для балки и груза. Для балки достаточно построить прямую линию, линия разбивается командой LMESH на заданные ранее балочные элементы. Для построения

квадрата используем команду RECTNG, которая строит прямоугольную область. Эта область разбивается на четырехугольные элементы PLANE42. Важно, что перед построением конечно-элементной модели нужно задать тип материала, тип конечного элемента и набор вещественных констант.

```
! Твердотельная и конечно-элементная модели балки
K,1,0,0
K,2,LBEAM,0
L,1,2
! Ассоциируем с балкой материал 1, тип КЭ 1, набор real
constantants 1
MAT,1 $ TYPE,1 $ REAL,1
ESIZE,,NE_BEAM ! Размер конечного элемента для балки
LMESH,ALL ! Разбиваем линию балки на элементы

! Твердотельная и конечно-элементная модели груза
RECTNG,L1,L1+AB,HB,HB+AB !Прямоугольник по двум точкам RECTNG,
X1, X2, Y1, Y2
! Ассоциируем с грузом материал 2, тип КЭ 2, набор real
constantants 2
MAT,2 $ TYPE,2 $ REAL,2
ESIZE,,NE_BLOCK ! Размер конечного элемента для груза
AMESH,ALL ! Разбиваем область на элементы
```

Для балки отобразим толщину элементов (пункты меню PlotCtrls->Style-> Size and shape->Display of element и выведем конечно-элементную модель. Она показана на рис. 1.



Рис. 2 Конечно-элементная сетка для балки и груза

Создадим контактные и ответные элементы. Контактные элементы будут на поверхности (границе) груза, а ответные – на балке. Контактные и ответные элементы создаются можно создать с помощью команды ESURF, которая генерирует элементы на свободных гранях существующих элементов. Ответные элементы сгенерируем на всей балке (предварительно нужно выбрать все элементы или узлы балки. Контактные элементы

создаются на границе груза, то есть на линиях, ограничивающих область груза.

! Создание (CONTA171) контактных и ответных (TARGE169) элементов

! Создание ответных элементов на балке

MAT,1 ! Созданные элементы будут принадлежать материалу 1
(балка)

TYPE,4 ! Задание типа элемента 4 (target)

REAL,3

NSEL,S,LOC,Y,0

TSHAP,LINE ! Форма ответного элемента - линия

CM,BEAM,NODE ! Выбранные узлы объединены в группу с именем beam

ESURF ! Генерирование ответных элементов на балке

ASEL,ALL !Выбор всех областей (груз)

LSLA,S ! Выбор линий на областях (граница груза)

nsll,s,1 ! Выбор узлов на выбранных линиях

TYPE,3 ! Задание типа элемента 3 (contact)

CM,PUNCH,NODE ! Выбранные узлы объединены в группу с именем
punch

ESURF ! Генерирование контактных элементов на поверхности груза

ALLSEL,ALL ! Выбор всего

Следующим этапом является задание граничных условий и решение задачи. Все эти действия выполним в решателе /SOLU.

Зададим тип анализа – нестационарный, а также эффекты нелинейной геометрии (опцию больших перемещений).

/SOLU

ANTYPE,TRANS ! Нестационарная задача

LUMPM,ON ! Диагональная матрица масс (не обязательная команда)

NLGEOM,ON ! Опция больших перемещений

Зададим условия закрепления для балки согласно условию задачи. Для груза следует жестко закрепить все его узлы. Зададим значение ускорения свободного падения.

! Условия закрепления для балки

DK,1,ALL,0

DK,2,UY,0

! Условия закрепления для груза

NSEL,S,LOC,Y,HB,HB+AB

D,ALL,ALL,0

NSEL,ALL

ACEL,,9.8 ! Учет ускорения свободного падения

Нестационарный анализ будет проведен в три этапа (три шага нагружения, load step). На первом шаге зададим очень небольшой временной интервал, чтобы рассчитать начальные условия для нестационарного анализа.

Решение получается в два подшага без интегрирования по времени. В файл записывается только последний результата.

```
! Первый шаг (load step)
! Расчет предварительного напряженного состояния
! Получение начальных условий для нестационарного анализа
TIME, TIME_0! Время в конце первого шага
DELTIM, DTIME_0! Шаг по времени (будет два подшага)
KBC, 1 ! Loads are step changed (stepped) at the first substep of
this load step
! to the values of this load step (i.e., the same values are used
for all substeps)
TIMINT, OFF ! No transient effects
outres, all, last ! Запись в файл результата последнего подшага
SOLVE ! Получение решения
```

Для перехода к следующему шагу следует удалить условия закрепления для груза.

```
! Выбор всех узлов груза
NSEL, S, LOC, Y, HB, HB+AB
DDELE, ALL, ALL ! Удаление условий закрепления
NSEL, ALL
```

Зададим опции нестационарного анализа для следующих шагов нагружения. Включим опцию интегрирования по времени, будем применять автоматический шаг по времени.

```
OUTRES, ALL, ALL ! Запись в файл результатов для всех подшагов
по времени
AUTOTS, ON ! Автоматический шаг по времени
TIMINT, ON ! Активизация нестационарных эффектов
```

Для второго интервала нагружения следует задать время окончания второго интервала и размер шага по времени.

```
! Второй шаг (load step)
TIME, TIME_1
DELTIM, DTIME_1, DTIME_1/5, DTIME_1*2 ! Шаг по времени, мин. и
макс. подшаги по времени
SOLVE
```



После получения решения зададим третий интервал. Длительность третьего интервала может быть произвольной. В данной задаче время окончания третьего интервала – это момент окончания наблюдения за грузом. Шаг по времени для третьего интервала должен быть меньше, чем для второго.

```
! Третий шаг (load step)
TIME, TIME_2
DELTIM, DTIME_2, DTIME_2/10, DTIME_2 ! Шаг по времени, мин. и макс. подшаги по
времени
SOLVE
```


АНАЛИЗ РЕЗУЛЬТАТОВ

Рассмотрим результаты как в общем постпроцессоре General Postproc (команда входа в общий постпроцессор: /POST1), так и во временном постпроцессоре TimeHist Postro (команда входа: /POST26). В общем постпроцессоре можно получить доступ к результатам, полученным для определенного момента времени. В постпроцессоре истории по времени TimeHist Postro можно получить доступ к результатам для определенного узла конечно-элементной модели для всего временного интервала. В частности, для заданного узла можно построить график зависимости определенной величины от времени.

Для начала построим графики перемещений узлов груза и балки во времени. Выведем конечно-элементную сетку (plot->Elements) и откроем Variable Viewer в TimeHist PostPro. Рассмотрим перемещения по оси y. Для этого добавим данные (зеленый плюс) – y-компоненту перемещения, задав имя переменной: block24.



Теперь укажем узел на грузе. Возьмем нижний узел посередине (номер 24). В список переменных добавится новая переменная:



Аналогично выберем узел для балки в месте проекции падения груза на балку (номер 14). Построим график вертикального смещения узла груза и балки (кнопка Graph Data), команды для подписей осей были даны в предыдущих лабораторных работах:

block24
beam14




Рис. 3 Графики перемещения U_y для узлов груза и балки

Выведем также отдельно график для смещения узла балки (номер 14):

beam14




Рис. 4 График перемещения U_y узла балки

Анимационный файл движения груза можно получить через пункты меню: PlotCtrls->Animate-> Over time

Animate Over Time

[ANTIME] Animate over time (interpolation of results)

Number of animation frames

Model result data

☐ Current Load Stp
☒ Load Step Range
☐ Time Range

Range Minimum, Maximum

Auto contour scaling ☒ On

Animation time delay (sec)

[PLDI,PLNS,PLVE,PLES,PLVFR] Contour data for animation

Use Last Display
DOF solution
Stress
Strain-total
Energy
Strain ener dens
Strain-elastic
Strain-thermal

Deformed Shape
Translation UX
UY
UZ
USUM
Rotation ROTX

Deformed Shape

OK Cancel Help